
Wish list for computational imaging code J. Fessler
CI in Julia

I Performance of a compiled language

JIT via LLVM

I Interactive development / prototyping

Dynamic typing / Jupyter notebooks

I Readability
◦ Syntax matching the mathematics of computational imaging

Unicode / UTF-8

◦ Built-in operations with numerical arrays

Yes

◦ Namespace control

using , import

I Relevant libraries (FFT, linear algebra, statistics, autograd, ...)

FFTW, LAPACK, ...

I Easy use of distributed computing, GPU resources

Yes

I Library ecosystem that facilitates reproducibility

git, Pkg, Manifest

I Free / open source

Yes

I IDE / debugger

Atom/Juno, VSCode, ...

I Code reuse (e.g., “object oriented”)

multiple dispatch, subtypes

I Memory efficient

Float16, Sparse of any type, call by ref.

I Interoperable with other languages

ccall, pycall, ...

Julia: incomplete history J. Fessler
CI in Julia

I https://julialang.org/
I Started in 2009 around MIT (Alan Edelman, Jeff Bezanson, Stefan Karpinski, and Viral B. Shah)

collaboration between computer sciences and computational sciences
I First release in 2012
I Version 1.0 in Aug. 2018
I 2017 SIAM Review paper [1]
I 2017 Forbes magazine article
I 2019 InfoWorld comparison of Julia and Python
I 2019 Nature article about Julia
I 2019 SIAM CSE Conference: James H. Wilkinson Prize for Numerical Software
I Sponsors of Juliacon 2018 and Juliacon 2019 include:

Microsoft, Amazon, Google, Intel, NVIDIA, CapitalOne, JP Morgan, ...
I Best of Matlab, Python, LISP, ...

https://julialang.org/
https://www.forbes.com/sites/suparnadutt/2017/09/20/this-startup-created-a-new-programming-language-now-used-by-the-worlds-biggest-companies/#9af992d7de2a
https://www.infoworld.com/article/3241107/python/julia-vs-python-julia-language-rises-for-data-science.html
http://doi.org/10.1038/d41586-019-02310-3
https://sinews.siam.org/Details-Page/scientific-machine-learning-how-julia-employs-differentiable-programming-to-do-it-best
http://juliacon.org/2018/
http://juliacon.org/2019/


UM history J. Fessler
CI in Julia

I Used at UM since 2017 for teaching
I EECS 551: Matrix methods in signal processing and machine learning
I EECS 505: Computational data science and machine learning
I EECS 598: Optimization methods for signal and image processing

I Michigan Image Reconstruction Toolbox (MIRT)
I Matlab version (inactive) https://github.com/JeffFessler/mirt
I Julia version (active) https://github.com/JeffFessler/MIRT.jl

I Group now uses mix of Matlab, python, Julia...

Syntax comparison J. Fessler
CI in Julia

Operation Matlab Julia Python import numpy as np

Dot product dot(x,y) dot(x,y) np.dot(x,y)
Matrix mult. A * B A * B A @ B
Element-wise A .* B A .* B A * B
Scaling 3 * A 3A or 3*A 3 * A
Matrix power A^2 A^2 np.linalg.matrix_power(A,2)
Element-wise A.^2 A.^2 A**2
Inverse inv(A) inv(A) np.linalg.inv(A)
Inverse A^(-1) A^(-1) np.linalg.inv(A)
Indexing A(i,j) A[i,j] A[i,j]
Range 1:9 1:9 np.arange(1,9,1)
Range linspace(0,4,9) LinRange(0,4,9) np.arange(0,4.01,0.5)
Strings ’text’ "text" (either)
Inline func. f = @(x,y) x+y f = (x,y) -> x+y f = lambda x,y : x+y
Increment A = A + B A += B A += B
Herm. transp. A’ A’ A.conj().T[
1 2
3 4

]
[1 2; 3 4] [1 2; 3 4] np.array([[1, 2], [3, 4]])

See https://cheatsheets.quantecon.org/ for more.

Kronecker sum example J. Fessler
CI in Julia

I Kronecker sum definition in math, for M ×M matrix A: and N × N matrix B:

A⊕ B , (IN ⊗ A) + (B ⊗ IM).

I Kronecker sum function in Julia:� �
using LinearAlgebra: I
⊗(A,B) = kron(A,B) # Kronecker product
⊕(A,B) = I(size(B,1)) ⊗ A + B ⊗ I(size(A)[1]) # Kronecker sum
X = ones(2,2) ⊕ [1 2; 3 4] # test drive� �

I using ... controls name-space
I f(x) = x^2 and f = x -> x^2 both define functions
I Enter symbols like ⊗ with LATEXcodes and tab completion:

� �
\otimes<tab>� �

I Julia supports argument chaining, e.g., size(A)[1]
7 / 34



Gradient descent example J. Fessler
CI in Julia

To minimize f (x) using gradient descent (for illustration) in math:

xk = xk−1 − α∇f (xk−1), for k = 1, 2, . . . ,Niter.

Gradient descent function code in Julia looks remarkably similar:� �
function gd(∇f::Function, α::Real, x ; Niter::Int = 50)

for iter=1:Niter
x = x - α * ∇f(x)

end
return x

end� �
8 / 34



Julia types and function arguments J. Fessler
CI in Julia� �

function gd(∇f::Function, α::Real, x ; Niter::Int = 50)� �
I Type annotation (via :: ) is optional (cf. x )
I Functions are first-class object types in Julia
I Julia is dynamically typed
I Real is an “abstract supertype” (step size α is any type of real number)
I subtypes(Real) returns

[AbstractFloat,AbstractIrrational,Integer,Rational]

I subtypes(AbstractFloat) returns [Float16,Float32,Float64,BigFloat]
(concrete number types: stored as bits)

I Optional named arguments (after ; ) with default values

9 / 34



Test GD routine J. Fessler
CI in Julia

f (x) = 1
2 ‖Ax − y‖22 =⇒ ∇f (x) = A′(Ax − y), L∇f = ‖A‖22 , α = 1/L∇f� �

using LinearAlgebra: opnorm # same as svdvals(A)[1]
include("gd.jl") # provides gd()
M,N = 6,4
A = randn(M,N); y = randn(M); # test data
∇f(x) = A'*(A*x - y) # LS gradient
xh = A \ y # global minimizer of f
α = 1/opnorm(A)ˆ2 # step size
xgd = gd(∇f, α, zeros(N) ; Niter=9000)
@assert xgd ≈ xh # equivalent within precision of type� �

I a ≈ b equivalent to isapprox(a,b)

Regularized LS (e.g., for computational imaging) J. Fessler
CI in Julia

x̂ = arg min
x

Ψ(x), Ψ(x) , 1
2 ‖Ax − y‖22 + R(x)

=⇒ ∇Ψ(x) = A′(Ax − y) +∇R(x)� �
include("gd.jl")
function rls(y, A, x, α::Real, ∇R::Function ; Niter::Int = 50)

∇Ψ(x) = A'*(A*x - y) + ∇R(x) # reg. LS cost gradient
gd(∇Ψ, α, x ; Niter=Niter) # returns updated x

end� �
I Could add type annotations like y::AbstractVector and A::AbstractArray

or A::Matrix{Float32}
I Using “duck typing” can facilitate code reuse
I Julia functions return the last quantity evaluated ( return x not required)

Regularized LS test J. Fessler
CI in Julia

R(x) = β
1
2 ‖x‖

2
2 =⇒ x̂ = arg min

x
Ψ(x) = (A′A + βI)−1A′y� �

using LinearAlgebra: I, opnorm # same as svdvals(A)[1]
include("rls.jl")
M,N = 6,4
A = randn(M,N); y = randn(M); # test data
β = 2; ∇R(x) = β*x # Tikhonov regularizer gradient
xh = (A'A + β*I) \ (A'*y) # global minimizer of Ψ
α = 1/(opnorm(A)ˆ2 + β) # step size
xr = rls(y, A, zeros(N), α, ∇R ; Niter=200)
@assert xr ≈ xh # equivalent within precision of type� �

Linear operators and overloading J. Fessler
CI in Julia

I Develop / debug code with A as a (small, dense) matrix
I For most computational imaging, A too big to store explicitly
I Want to avoid rewriting new versions of GD and RLS for every A
I Apply code with more general linear operator A that can do A*x and A’*y
I Minimalist version of “operator overloading:”� �

struct LinOp # see LinearMaps.jl and LinearMapsAA.jl for pro version
forward::Function # A * x will call forward(x)
adjoint::Function # A' * y will call adjoint(y)

end

Base.:(*)(A::LinOp, x) = A.forward(x) # for A*x
Base.adjoint(A::LinOp) = LinOp(A.adjoint, A.forward) # for A'� �

I Full-scale version https://github.com/JeffFessler/LinearMapsAA.jl

Multiple dispatch introduction J. Fessler
CI in Julia� �

struct LinOp # see LinearMaps.jl and LinearMapsAA.jl for pro version
forward::Function # A * x will call forward(x)
adjoint::Function # A' * y will call adjoint(y)

end

Base.:(*)(A::LinOp, x) = A.forward(x) # for A*x
Base.adjoint(A::LinOp) = LinOp(A.adjoint, A.forward) # for A'� �
Works because of Julia’s multiple dispatch feature
I Infix x * y is just syntax for *(x,y)

I By length(methods(*)) , there are over 350 instances of *() in Julia 1.2, for
different combinations of argument types

I A’ is just syntax for adjoint(A) that has about 40 pre-defined methods.
I The code above “overloads” adjoint for a new argument type, enabling A’

No need for syntax like A.adjoint(), enabling reuse of matrix-like code
14 / 34



Multiple dispatch continued J. Fessler
CI in Julia

I Many computational imaging system models involve products of linear operators
◦ super-resolution forward model has blur and down-sampling
◦ compressed sensing MRI has coil sensitivity, Fourier transform, sampling

I Consider A = BC for which Ax = B(Cx) and A′y = C ′(B′y)� �
Base.:(*)(A::LinOp, x) = A.forward(x) # for A*x
Base.:(*)(B::LinOp, C::LinOp) = # for B*C (composition)

LinOp(B.forward ◦ C.forward, C.adjoint ◦ B.adjoint)� �
I Multiple dispatch depends on all argument types (not just first argument)
I Julia compiler chooses most specific version of function
I The ◦ above is function composition.

Finite-difference regularizer in 2D J. Fessler
CI in Julia

2D finite difference regularizer with periodic boundary conditions in sum form:

R(x) = β
N∑

n=1

M∑
m=1

φ(x [n,m]− x [n − 1 mod N,m]) + φ(x [n,m]− x [n,m − 1 mod M])

=
MN∑
k=1

φ([T1x]k) + φ([T2x]k) =
2MN∑
k=1

φ([Tx]k), T ,

[
T1
T2

]
= 1′φ.(Tx),

where φ.(v) denotes applying potential function φ element-wise to v :

φ(v1)
φ(v2)
...

 .
If φ(t) = |t| then 1′φ.(Tx) = ‖Tx‖1 .

Finite-difference regularizer in 4 lines of Julia J. Fessler
CI in Julia� �

include("linop.jl")

diff2d_forw = x -> cat(x - circshift(x, (0,1)), x - circshift(x, (1,0)), dims=3)
diff2d_back = y -> (y[:,:,1] - circshift(y[:,:,1], (0,-1))) +

(y[:,:,2] - circshift(y[:,:,2], (-1,0)))
T = LinOp(diff2d_forw, diff2d_back)

M,N = 6,8 # testing
x = randn(M,N)
y = randn(M,N,2)
@assert sum(y .* (T * x)) ≈ sum(x .* (T' * y)) # check adjoint� �
I In math: T1 = I − P1 =⇒ T1x = x − P1x,

where P1 denotes the circular shift along first index
I Here the resulting linear operator T maps a 2D array into a 3D array.
I T’ is the adjoint (not transpose) of that operator.
I .* does element-wise multiplication

Regularizer gradient J. Fessler
CI in Julia

Gradient of regularizer uses derivative φ̇ of potential function element-wise:

R(x) =
∑
k
φ([Tx]k) = 1′φ.(Tx) =⇒ ∇R(x) = T ′φ̇.(Tx).

Example: Fair potential function φ̇(z) = β
z

1 + |z | /δ� �
δ = 0.1 # parameter for Fair potential function
β = 5 # regularization parameter
dϕ(z) = β * z / (1 + abs(z)/δ) # derivative of potential function
∇R(x) = T' * dϕ.(T * x) # regularizer gradient� �
I f.(v) applies f element-wise in Julia
I Julia matches math or vice-versa?

Image denoising example J. Fessler
CI in Julia

y = x + ε so A = I. Use x0 = y� �
include("rls.jl"); include("diff2d.jl") # makes T
using LinearAlgebra: I
using MIRT: jim, ellipse_im
using Random: seed!
using Plots: plot
M,N = 120,128; xtrue = ellipse_im(M,N) # shepp-logan digital phantom
σ = 0.2; seed!(0); y = xtrue + σ * randn(size(xtrue)) # noisy data

δ = 0.1 # parameter for Fair potential function
β = 5; dϕ(z) = β * z / (1 + abs(z)/δ) # derivative of potential function
∇R(x) = T' * dϕ.(T * x) # regularizer gradient
α = 1 / (1 + 8β) # step size = 1 / Lipschitz constant
xh = rls(y, I, y, α, ∇R ; Niter=60)
plot(jim(y, title="Noisy", clim=[0,8]), # jiffy image display

jim(xh, title="Denoised", clim=[0,8])) # savefig("denoise-y-xh.pdf")� �

Image denoising result J. Fessler
CI in Julia

Super-resolution system model J. Fessler
CI in Julia

Simple 2×-downsampling model:
y [m, n] = x [2m, 2n] + ε[m, n] corresponds to y = Ax + ε for A : R2M×2N 7→ RM×N� �
include("linop.jl") # defines LinOp
down2_forw = (x) -> x[1:2:end,1:2:end] # down-sampling in 1 line
function down2_adjoint(y) # up-sampling (adjoint) takes 2 lines:

x = zeros(eltype(y), size(y).*2)
x[1:2:end,1:2:end] .= y # key step for up-sampling
return x

end
A = LinOp(down2_forw, down2_adjoint)

M,N = 60,64; x = randn(2M,2N); y = randn(M,N) # test data
@assert sum(y .* (A * x)) ≈ sum(x .* (A' * y)) # check adjoint� �
I .= does element-wise (in-place) assignment

Super-resolution example J. Fessler
CI in Julia� �

include("rls.jl"); include("diff2d.jl") # makes T
include("down2-linop.jl") # makes A
using MIRT: jim, ellipse_im
using Random: seed!
using Plots: plot, savefig
using ImageFiltering: imfilter
M,N = 120,128; xtrue = ellipse_im(M,N) # shepp-logan digital phantom
σ = 0.1; seed!(0); y = A*xtrue; y += σ * randn(size(y)) # noisy data

x0 = imfilter(A'*y, ones(2,2)) # initial guess
δ = 0.1 # parameter for Fair potential function
β = 2; dϕ(z) = β * z / (1 + abs(z)/δ) # derivative of potential function
∇R(x) = T' * dϕ.(T * x) # regularizer gradient
α = 1 / (1 + 8β) # 1 / Lipschitz constant (‖A‖2 = 1)
xh = rls(y, A, x0, α, ∇R ; Niter=600) # same RLS as before!
plot(jim(y, title="Noisy Lo-Res data", clim=[0,8]),

jim(xh, title="Recovered", clim=[0,8])) # savefig("superres-y-xh.pdf")� �
22 / 34



Super-resolution results J. Fessler
CI in Julia

Training a very basic NN using ForwardDiff.jl J. Fessler
CI in Julia� �

relu = x -> x < 0 ? 0.1*x : x # leaky ReLU
neuron1(W::Matrix, b, x) = relu.(W * x + b) # perceptron: weights, bias, nonlin
neuron1(w::Vector, b, x) = neuron1(reshape(w, length(b), length(x)), b, x)

include("un_vcat.jl")
include("gd.jl")
using LinearAlgebra: norm
using Random: seed!
using ForwardDiff

seed!(0)
Ntrain = 100
xtrain = sort((rand(Ntrain) .- 0.5) * 4)
ftrue = x -> min(xˆ2,1) # broken parabola
ytrain = ftrue.(xtrain) + 0.01 * randn(Ntrain) # training data

Nhidden = 10 # 1 hidden layer
Nparams = [Nhidden,1,Nhidden,Nhidden] # (w2,b2,w1,b1)
Nparam = sum(Nparams)
layer1 = (w1,b1) -> (x -> neuron1(w1, b1, x)) # NN architecture
layer2 = (w2,b2) -> (x -> neuron1(w2, b2, x)[1]) # scalar output
predict = (w2,b2,w1,b1) -> layer2(w2,b2) ◦ layer1(w1,b1) # compose
predict1 = w -> predict(un_vcat(w, Nparams)...) # tricky w = vcat(w2,b2,w1,b1)
loss1 = w -> norm(ytrain - predict1(w).(xtrain))ˆ2
∇loss1 = w -> ForwardDiff.gradient(loss1, w) # so easy!

w0 = 0.2 * randn(Nparam) # initial guess of NN weights
α = 1e-3 # step size = learning rate
wh = gd(∇loss1, α, w0 ; Niter = 9000); @show loss1(w0), loss1(wh)� � 24 / 34



One helper function J. Fessler
CI in Julia� �

"""
undo the effect of `vcat(xs...)`
"""
function un_vcat(x::AbstractVector{T}, dim::AbstractVector{Int}) where {T}

N = length(dim)
xs = Array{AbstractVector{T}}(undef, N)
for n=1:N

i0 = sum(dim[1:(n-1)])
xs[n] = x[(i0+1) : (i0 + dim[n])]

end
return xs

end

using Test: @test, @inferred
x1 = 1:4; x2 = ones(5); x3 = [7,6]
dim = [length(x1), length(x2), length(x3)]
x = vcat(x1, x2, x3)
xs = @inferred un_vcat(x, dim)
@test xs == [x1, x2, x3]� �

Key to automatic differentiation is Julia’s metaprogramming (cf. LISP)
& reflection & introspection

https://en.wikipedia.org/wiki/Reflection_(computer_programming)
https://en.wikipedia.org/wiki/Type_introspection

I Many plotting back-ends including pyplot
◦ Interact package provides GUI elements (sliders, buttons) in Jupyter
◦ Time-to-first-plot needs improvement

I No clear function
x = nothing frees memory used by x

I Julia default indexing is 1-based, not 0-based
◦ for x in list

for (i,x) in enumerate(list)
◦ Can create variables with other indexing conventions
◦ FFTViews package enables [−N/2, . . . ,N/2− 1] indexing for DSP
◦ Array comprehension largely eliminates need for ndgrid

[function(j,k,l) for j=0:J-1, k=0:0.5:K-1, l=1:L]
I Encapsulation can be circumvented
I Default is 1 thread (config file option to multithread)
I Obsolete Q/A on stackexchange etc.

https://en.wikipedia.org/wiki/Encapsulation_(computer_programming)


Random notes J. Fessler

I Detailed documentation at https://docs.julialang.org
Online tutorials, stackexchange, github conversations, wikibook:
https://en.wikibooks.org/wiki/Introducing_Julia

I 7:0.1:100000 stored as Range type, not as a huge array
I Julia does not force you to indent any particular way
I Functions can return multiple arguments� �

rotate = (x,y,θ) -> (x*cos(θ) + y*sin(θ), -x*sin(θ) + y*cos(θ)) # 3 in, 2 out
xnew, ynew = rotate(x, y, π/4) # use both outputs� �

https://docs.julialang.org
https://en.wikibooks.org/wiki/Introducing_Julia

"""
This function applies `Niter` iterations of GD
to compute the minimizer ``\\hat{x}`` of the cost function
``\\Psi(x) = \\frac{1}{2} \\| A x - y \\|_2^2 + R(x)``
given the gradient ``\\nabla R``
"""
function rls(y, A, x, α::Real, ∇R::Function ; Niter::Int = 50)

∇Ψ(x) = A'*(A*x - y) + ∇R(x) # reg. LS cost gradient
gd(∇Ψ, α, x ; Niter=Niter) # returns updated x

end� �

Multiple dispatch comparison J. Fessler

From 2019 JuliaCon talk by Stefan Karpinski https://youtu.be/kc9HwsxE1OY?t=679

https://youtu.be/kc9HwsxE1OY?t=679


Juno/Atom in action / python integration J. Fessler

• Python integration
• @

• Debugger feedback
From 2019 JuliaCon talk
by Mike Innes
https://youtu.be/OcUXjk7DFvU?t=1055

https://youtu.be/OcUXjk7DFvU?t=1055


Further reading J. Fessler

I Oct. 2019 SIAM News article ablut differentiable programming in Julia
I Forward-mode automatic differentiation [2] with ForwardDiff.jl
I Reverse-mode automatic differentiation [3] with Zygote.jl

I Compilation to cloud TPUs [4]
I Optimization package(s) [5]
I GPU integration [6]
I Intro to machine learning with Julia

https://tinyurl.com/ml2-18-jf
I Help wanted!

https://github.com/JeffFessler/MIRT.jl/blob/master/doc/matlab-to-julia.md

Acknowledgment: Beamer slides made with modified jlcode.sty from
https://github.com/wg030/jlcode

https://sinews.siam.org/Details-Page/scientific-machine-learning-how-julia-employs-differentiable-programming-to-do-it-best
https://github.com/JuliaDiff/ForwardDiff.jl
https://github.com/FluxML/Zygote.jl
https://tinyurl.com/ml2-18-jf
https://github.com/JeffFessler/MIRT.jl/blob/master/doc/matlab-to-julia.md
https://github.com/wg030/jlcode

Talk and code available online at
http://web.eecs.umich.edu/~fessler/papers/files/talk/19/ima-julia

